This Maple worksheet accompanies the paper

Wolfram Koepf, Dieter Schmersau: Positivity and Monotony Properties of the de Branges Functions (2003)

and contains the Maple computations for some of the theorems in the paper. Details on the algorithms used can be found in the book

Wolfram Koepf: Hypergeometric Summation. Vieweg, Braunschweig/Wiesbaden, 1998.

```
> restart;
> with(sumtools);
```

$$[Hypersum, Sumtohyper, extended\_gosper, gosper, hyperrecursion, hypersum, hyperterm, simpcomb, sumrecursion, sumtohyper]$$

```
> tauterm:=y^k*binomial(n+k+1,2*k+1)*hyperterm([k+1/2,n+k+2,k,k-n],[k+1,2*k+1,k+3/2],y,j);
```

$$tauterm := y^k \, \mathrm{binomial}(n+k+1, 2k+1)\, \mathrm{pochhammer}\left(k+\frac{1}{2}, j\right) \mathrm{pochhammer}(n+k+2, j)\, \mathrm{pochhammer}(k, j)\, \mathrm{pochhammer}(k-n, j)\, y^j \Big/ \left(\mathrm{pochhammer}(k+1, j)\, \mathrm{pochhammer}(2k+1, j)\, \mathrm{pochhammer}\left(k+\frac{3}{2}, j\right) j!\right)$$

```
> Lambdaterm:=y^k*binomial(n+k+1,2*k+1)*hyperterm([k+1/2,n+k+2,k-n],[2*k+1,k+3/2],y,j);
```

$$Lambdaterm := y^k \, \mathrm{binomial}(n+k+1, 2k+1)\, \mathrm{pochhammer}\left(k+\frac{1}{2}, j\right) \mathrm{pochhammer}(n+k+2, j)\, \mathrm{pochhammer}(k-n, j)\, y^j \Big/ \left(\mathrm{pochhammer}(2k+1, j)\, \mathrm{pochhammer}\left(k+\frac{3}{2}, j\right) j!\right)$$

The next computations use Zeilberger’s algorithm to deduce the recurrence equations presented in Theorem 6.

```
> RE1:=sumrecursion(tauterm,j,tau(n));
```

$$\begin{aligned} RE1 := &(2n-1)(n+k-2)(n-k-2)\,\tau(n-4) \\ &+ 2(4yn^3 - 4n^3 + 15n^2 - 12yn^2 + 11yn - 17n + 5 + k^2 - 3y)\,\tau(n-3) \\ &- 4(n-1)(4yn^2 - 3n^2 - 8yn + 6n - k^2 - 2 + 3y)\,\tau(n-2) \\ &+ 2(4yn^3 - 4n^3 - 12yn^2 + 9n^2 - 5n + 11yn - 3y - k^2 + 1)\,\tau(n-1) \\ &+ (-3+2n)(n-k)(n+k)\,\tau(n) \end{aligned}$$

```
> # very time and memory consuming!
TIME:=time():
RE2:=sumrecursion(tauterm,j,tau(k));
time()-TIME;
```

$$RE2 := y\,(k-1)\,(2\,k-3)\,(n+k-2)\,(n-k+4)\,\tau(k-4)-2\,(k-1)$$
$$(4\,k^3-4\,y\,k^3+29\,y\,k^2-28\,k^2-67\,y\,k+63\,k+6\,y\,n+3\,y\,n^2+51\,y-45)\,\tau(k-3)$$
$$+2\,(k-3)$$
$$(4\,y\,k^3-4\,k^3-19\,y\,k^2+20\,k^2+27\,y\,k-31\,k+6\,y\,n+3\,y\,n^2-9\,y+15)\,\tau(k-1)$$
$$+y\,(2\,k-5)\,(k-3)\,(n+k)\,(n-k+2)\,\tau(k)-2\,(2\,y\,k^2\,n^2-8\,y\,k\,n^2+3\,y\,n^2$$
$$+4\,y\,k^2\,n-16\,y\,k\,n+6\,y\,n+6\,y\,k^4-8\,k^4-48\,y\,k^3+64\,k^3+137\,y\,k^2-182\,k^2$$
$$-164\,y\,k+216\,k+66\,y-90)\,\tau(k-2)$$

44180.710

```
> RE3:=sumrecursion(Lambdaterm,j,Lambda(n));
```

$$RE3 := -(n+k-1)\,(n-k-1)\,n\,\Lambda(n-3)$$
$$-(-3\,n^2+4\,y\,n^2-2\,y\,n+2\,n-k^2)\,\Lambda(n-2)\,(n-1)$$
$$+n\,(-3\,n^2+4\,n+4\,y\,n^2-6\,y\,n+2\,y-1-k^2)\,\Lambda(n-1)$$
$$+(n-k)\,(n+k)\,(n-1)\,\Lambda(n)$$

```
> TIME:=time():
RE4:=sumrecursion(Lambdaterm,j,Lambda(k));
time()-TIME;
```

$$RE4 := -y\,(k-1)\,(n+k-1)\,(n+3-k)\,\Lambda(k-3)$$
$$-(k-2)\,(3\,y\,k^2-4\,k^2-8\,y\,k+10\,k+y\,n^2+2\,y\,n+6\,y-6)\,\Lambda(k-2)$$
$$+(k-1)\,(3\,y\,k^2-4\,k^2-10\,y\,k+14\,k+y\,n^2+9\,y-12+2\,y\,n)\,\Lambda(k-1)$$
$$+y\,(k-2)\,(n+k)\,(n-k+2)\,\Lambda(k)$$

28.552

The following yields the hypergeometric representations that are used in the proof of Theorem 8, i.e. Eq. (22).

```
> df:=sumtohyper(tauterm-subs(n=n-1,tauterm),j);
```

$$df := y^k\,(\mathrm{binomial}(n+k+1,2\,k+1)-\mathrm{binomial}(n+k,2\,k+1))$$
$$\mathrm{hypergeom}([k,k-n,n+k+1],[2\,k+1,k+1],y)$$

```
> prefactor:=eval(df,hypergeom=1);
```

$$prefactor := y^k\,(\mathrm{binomial}(n+k+1,2\,k+1)-\mathrm{binomial}(n+k,2\,k+1))$$

```
> simpcomb(y^k*binomial(n+k,n-k)/prefactor);
```

1

Theorem 9:

```
> s:=sumtohyper(tauterm+Lambdaterm,j);
```

$$s := 2\,y^k\,\mathrm{binomial}(n+k+1,2\,k+1)$$
$$\mathrm{hypergeom}\left(\left[k,k+\frac{1}{2},k-n,n+k+2\right],\left[k+1,k+\frac{3}{2},2\,k\right],y\right)$$

```
> d:=sumtohyper(tauterm-Lambdaterm,j);
```

$$d := (n-k)\,(n+k+2)\,\mathrm{binomial}(n+k+1,\,2\,k+1)\,y^{(k+1)}\,\mathrm{hypergeom}\left(\left[n+k+3,\,k+1,\,k+\frac{3}{2},\,k-n+1\right],\,\left[k+2,\,k+\frac{5}{2},\,2\,k+2\right],\,y\right)/((2\,k+3)\,(k+1))$$

```
> simpcomb(simplify(subs(k=k+1,s)/d));
```

$$1$$

```
>
```